Excel COUNTIF Function



Summary

COUNTIF is an Excel function to count cells in a range that meet a single condition. COUNTIF can be used to count cells that contain dates, numbers, and text. The criteria used in COUNTIF supports logical operators (>,<,<>,=) and wildcards (*,?) for partial matching.

Purpose

Count cells that match criteria

Return value

A number representing cells counted.

Syntax

=COUNTIF (range, criteria)

Arguments

range - The range of cells to count.

criteria - The criteria that controls which cells should be counted.

Version

Excel 2003

Usage notes

The COUNTIF function in Excel counts the number of cells in a range that match one supplied condition. Criteria can include logical operators (>,<,<>,=) and wildcards (*,?) for partial matching. Criteria can also be based on a value from another cell, as explained below.

COUNTIF is in a group of eight functions in Excel that split logical criteria into two parts (range + criteria). As a result, the syntax used to construct criteria is different, and COUNTIF *requires* a cell range, you can't use an array.

COUNTIF only supports a *single* condition. If you need to apply multiple criteria, use the COUNTIFS function. If you need to manipulate values in the *range* argument as part of a logical test, see the SUMPRODUCT and/or FILTER functions.

Basic example

In the worksheet shown above, the following formulas are used in cells G5, G6, and G7:

```
= COUNTIF(D5:D12,">100") // count sales over 100
= COUNTIF(B5:B12,"jim") // count name = "jim"
= COUNTIF(C5:C12,"ca") // count state = "ca"
```

Notice COUNTIF is *not* case-sensitive, "CA" and "ca" are treated the same.

Double quotes ("") in criteria

In general, text values need to be enclosed in double quotes (""), and numbers do not. However, when a logical operator is included with a number, the number and operator must be enclosed in quotes, as seen in the second example below:

```
= COUNTIF(A1:A10,100) // count cells equal to 100
= COUNTIF(A1:A10,">32") // count cells greater than 32
= COUNTIF(A1:A10,"jim") // count cells equal to "jim"
```

Value from another cell

A value from another cell can be included in criteria using concatenation. In the example below, COUNTIF will return the count of values in A1:A10 that are less than the value in cell B1. Notice the less than operator (which is text) is enclosed in quotes.

```
= COUNTIF(A1:A10,"<" & B1) // count cells less than  B1
```

Not equal to

To construct "not equal to" criteria, use the "<>" operator surrounded by double quotes (""). For example, the formula below will count cells *not equal* to "red" in the range A1:A10:

```
= COUNTIF(A1:A10,"<>red") // not "red"
```

Blank cells

COUNTIF can count cells that are blank or not blank. The formulas below count blank and not blank cells in the range A1:A10:

```
= COUNTIF(A1:A10,"<>") // not blank
= COUNTIF(A1:A10,"") // blank
```

Dates

The easiest way to use COUNTIF with dates is to refer to a valid date in another cell with a cell reference. For example, to count cells in A1:A10 that contain a date greater than the date in B1, you can use a formula like this:

```
= COUNTIF(A1:A10, ">" & B1) // count dates greater than A1
```

Notice we must concatenate an operator to the date in B1. To use more advanced date criteria (i.e. all dates in a given month, or all dates between two dates) you'll want to switch to the COUNTIFS function, which can handle multiple criteria.

The safest way *hardcode* a date into COUNTIF is to use the DATE function. This ensures Excel will understand the date. To count cells in A1:A10 that contain a date less than April 1, 2020, you can use a formula like this

```
= COUNTIF(A1:A10,"<" & DATE(2020,4,1)) // dates less than 1-Apr-2020
```

Wildcards

The wildcard characters question mark (?), asterisk(*), or tilde (~) can be used in criteria. A question mark (?) matches any one character and an asterisk (*) matches zero or more characters of any kind. For example, to count cells in a A1:A5 that contain the text "apple" anywhere, you can use a formula like this:

```
= COUNTIF(A1:A5,"*apple*") // cells that contain "apple"
```

To count cells in A1:A5 that contain any 3 text characters, you can use:

```
= COUNTIF(A1:A5,"???") // cells that contain any 3 characters
```

The tilde (~) is an escape character to match literal wildcards. For example, to count a literal question mark (?), asterisk(*), or tilde (~), add a tilde in front of the wildcard (i.e.

~?, ~*, ~~).

Notes

- COUNTIF is not case-sensitive. Use the EXACT function for case-sensitive counts.
- COUNTIF only supports one condition. Use the COUNTIFS function for multiple criteria.
- Text strings in criteria must be enclosed in double quotes (""), i.e. "apple", ">32", "ja*"
- Cell references in criteria are *not* enclosed in quotes, i.e. "<"&A1
- The wildcard characters ? and * can be used in criteria. A question mark matches any one character and an asterisk matches any sequence of characters (zero or more).
- To match a literal question mark or asterisk, use a tilde (~) in front question mark or asterisk (i.e. ~?, ~*).
- COUNTIF *requires* a range, you can't substitute an array.
- COUNTIF returns incorrect results when used to match strings longer than 255 characters.
- COUNTIF will return a #VALUE error when referencing another workbook that is closed.

COUNTIF formula examples



Count unique numeric values in a range

Note: Prior to Excel 365, Excel did not have a dedicated function to count unique values. This formula shows a one way to count unique values, as long as they are numeric. If you have text values, or a mix of text and...



Unique values by count

This example uses the UNIQUE function together with the FILTER function. You can see a more basic example here. The trick in this case is to apply criteria to the FILTER function to only allow values based on count of...



Flag first duplicate in a list

At the core, this formula is composed of two sets of the COUNTIF function wrapped in the IF function. The outer IF + COUNTIF first checks to see if the the value in question (B4) appears more than once in the list: =...



if cell contains this or that

The core of this formula is COUNTIF, which returns zero if none of the substrings is found, and a positive number if at least one substring is found. The twist in this case is that we are giving COUNTIF more than one...



Count cells that end with

COUNTIF counts the number of cells in the range that end with "r" by matching the content of each cell against the pattern "*r", which is supplied as the criteria. The "*" symbol (the asterisk) is a wildcard in Excel...



Student class enrollment with table

Note the purpose of this example is to how one way to "normalize" data when the order of values is random. There are many ways to approach this problem. The formula in G6 relies on the COUNTIF function to count the...



Range contains one of many substrings

All the hard work is done by the COUNTIF function, which is configured to count the values in the named range "substrings" that appear the named range "rng" with like this: COUNTIF(rng,"*"&substrings...



Count cells that contain text

COUNTIF counts the number of cells that match the supplied criteria. In this case, the criteria is supplied as the wildcard character "*" which matches any number of text characters. A few notes: The logical values...



Count cells less than

COUNTIF counts the number of cells in the range that contain numeric values less than X and returns the result as a number. If you want to count cells that are "less than or equal to 80", use: =COUNTIF(C5:C11,...



Running count of occurrence in list

Working from the inside out, the COUNTIF function is set up to count the value "blue" that appears in column B: COUNTIF(B5:B5,"blue") Note the left side of the range reference is locked (B5) and the...



Count cells not equal to

In Excel, the operator for not equal is "". For example: =A1<>10 // A1 is not equal to 10 =A1<>"apple" // A1 is not equal to "apple" The COUNTIF function counts the number of cells in...



Volunteer hours requirement calculation

There are two requirements that must be satisfied, and both evaluated inside a single AND statement. The first requirement is at least 5 volunteer hours in each of the 3 categories: A, B, and C. This requirement is...



Get nth match with VLOOKUP

This formula depends on a helper column, which is added as the first column to the source data table. The helper column contains a formula which builds a unique lookup value from an existing id and a counter. The...



Multiple columns are equal

In the example shown, we want to test if all values in each row are equal. To do this, we use an expression that compares the value in the first column (B5) to the rest of the columns (C5:F5): B5=C5:F5 Because we...



Data validation must begin with

Data validation rules are triggered when a user adds or changes a cell value. In this formula, the LEFT function is used to extract the first 3 characters of the input in C5. Next, the EXACT function is used to...

Related videos



How to use the COUNTIF function

In this video, we'll look at how to use the COUNTIF function to count cells that meet a multiple criteria in a set of data.



How to create a dynamic named range with a Table

In this video, we'll look at how to create a dynamic named range with a Table. This is the simplest way to create a dynamic named range in Excel.

How to plot survey data in a bar chart

In this video, we'll look at how to plot over 3000 survey responses to a question in an Excel bar chart.

How to find missing values with COUNTIF

In this video, we'll take a look at how to use the COUNTIF function to solve a common problem: how to find values in one list that appear in another list.

Formulas to query a table

Because tables support structured references, you can learn a lot about a table with simple formulas. In this video, we'll look at some formulas you can use to query a table.



List duplicate values with FILTER

In this video, we'll look at how to list duplicate values with the FILTER function. In other words, values that appear more than once in a set of data.

Related functions



Excel COUNTIFS Function

The Excel COUNTIFS function returns the count of cells that meet one or more criteria. COUNTIFS can be used with criteria based on dates, numbers, text, and other conditions. COUNTIFS supports logical operators (>,...

See also

Excel's RACON functions

Topics

- Formula Basics
- Formula Examples
- Conditional Formatting
- Pivot Tables
- Excel Tables
- Excel Charts
- Shortcuts
- More...

Key functions

- IF function
- VLOOKUP function
- XLOOKUP function
- FILTER function
- SUMIFS function
- COUNTIFS function
- SUMPRODUCT function
- INDEX and MATCH
- More functions...

Hi - I'm Dave Bruns, and I run Exceljet with my wife, Lisa. Our goal i
help you work faster in Excel. We create short videos, and
clear examples of formulas, functions, pivot tables,
conditional formatting, and charts. Read more.

Just wanted to say thanks! Your explanations are very concise and thoughtful. I just searched for conditional formatting based on values on a list. Your idea blew my mind. It seemed so simple and creative! Awesome! It saved me creating a macro for that. I will look at excel formulas from a different angle now. -Ricardo